

▼ Aufgabe aus der Analysis

ÜBUNG [01]:

Untersuchen Sie, wieviele reelle Lösungen die Gleichung $x(x - \sin(x)) = \cos(x)$ hat.

▼ Differentiation: Definition, Eigenschaften und Beispiele

[Aufgaben: 3

[> **restart**;

▼ Definition Differentiation

Differenzieren ist eines der zentralen Hilfsmittel zur Untersuchung der Wachstumseigenschaften reeller Funktionen. Die meisten gängigen Funktionen sind differenzierbar, aber leider nicht alle.

MATH: Wir betrachten Funktionen $f: \mathbb{R} \rightarrow \mathbb{R}$ die in $x_0 \in \mathbb{R}$ stetig sind. Stetigkeit in x_0 kann man auch so sehen, dass f in der Nähe von x_0 durch die konstante Funktion

$$v_0: \mathbb{R} \rightarrow \mathbb{R}: x \mapsto f(x_0)$$


gut angenähert wird. Wenn dies eintritt, kann man mit etwas allgemeineren, aber immer noch sehr eingeschränkten Funktionen

$$v_1: \mathbb{R} \rightarrow \mathbb{R}: x \mapsto f(x_0) + a \cdot (x - x_0)$$

mit $a \in \mathbb{R}$ vergleichen, denn diese sind auch stetig in x_0 und haben dort auch den Wert $f(x_0)$.

BEISPIEL: Hier wird $f: x \mapsto x^2$ bei $x_0 = 1$ mit zwei verschiedenen v_1 verglichen: eine (grün) mit $a = 2$ und eine (rot) mit $a = 1$. Warum ist $a = 2$ besser?

```
> plot(subs(x0=1,[(x0+h)^2, x0^2+2*x0*h, x0^2+h]),h=-1/2..1/2,  
color=[black,green,red],thickness=2);
```



MATH: Alle $f - v_1$ sind stetig in x_0 mit dem Wert 0, ebenso wie

$V: \mathbb{R} \rightarrow \mathbb{R} : x \mapsto x - x_0$. Also bietet es sich an, nach der Stetigkeit des Quotienten

$$\frac{f - v_1}{V} : x \mapsto \frac{f(x) - f(x_0) - a \cdot (x - x_0)}{x - x_0} = \frac{f(x) - f(x_0)}{x - x_0} - a$$

zu fragen, der auf $\mathbb{R} - \{x_0\}$ definiert ist. Zwei Möglichkeiten ergeben sich:

$\frac{f - v_1}{V}$ ist stetig ergänzbar in x_0 oder nicht. Im ersten Fall heißt

f **differenzierbar** in x_0 . Verlangt man noch in Analogie zur Stetigkeit, wo $(f - v_0)(x_0) = 0$ gilt, dass der ergänzte Funktionswert gleich Null ist, so tritt dies für genau einen Wert von a ein und man sagt: f ist **differenzierbar** in x_0 mit **Ableitung** $f'(x_0) = a$.

Beachte: In diesem Fall ist v_1 mit dieser Wahl von a eine wesentlich bessere Annäherung an f in x_0 als v_0 , d. h. für x gegen x_0 konvergiert $(f - v_1)(x)$ schneller gegen Null als $(f - v_0)(x)$. Im zweiten Fall, wo $\frac{f - v_1}{V}$ in x_0 nicht stetig ergänzbar ist, sagt man, f ist nicht differenzierbar in x_0 .

$$> \lim_{h \rightarrow 0} \frac{(x_0 + h)^2 - x_0^2}{h} = 2x_0 \quad (2.1.1)$$

$$> \lim_{x \rightarrow x_0} \frac{x^2 - x_0^2}{x - x_0} = 2x_0 \quad (2.1.2)$$

Also ist in diesem Fall $\frac{f - v_1}{V}$

$$> \frac{x^2 - (x_0^2 + 2x_0(x - x_0))}{x - x_0};$$

$$\frac{x^2 - x_0^2 - 2x_0(x - x_0)}{x - x_0} \quad (2.1.3)$$

Diese lässt sich in der Tat stetig in x_0 (als Funktion von x) ergänzen, denn

```
> simplify(%);
```

$$x - x_0 \quad (2.1.4)$$

MAPLE: Maple geht zunächst einmal davon aus, dass eine beliebig vorgegebene (unspezifizierte) Funktion differenzierbar ist und hat seine eigene Notation für die Ableitung:

```
> limit((f(x)-f(x0))/(x-x0), x=x0);
```

$$D(f)(x_0) \quad (2.1.5)$$

MAPLE: Im Falle konkreter, nicht differenzierbarer Funktionen protestiert Maple:

```
> limit((abs(x)-abs(0))/(x-0), x=0);
```

$$\text{undefined} \quad (2.1.6)$$

MATH: Differenzierbarkeit in einem Punkt ist weniger interessant als Differenzierbarkeit auf dem ganzen Definitionsbereich, z. B. auf \mathbb{R} oder auf einem offenen Intervall in \mathbb{R} . Ist der Fall gegeben, so kann man die Ableitung wieder als Funktion auf dem Definitionsbereich auffassen.

MAPLE: Für die gängigen Fälle kann MAPLE symbolisch differenzieren:

```
> diff(x^2, x);
```

$$2x \quad (2.1.7)$$

Gemeint ist, die Ableitung der Funktion $\mathbb{R} \rightarrow \mathbb{R} : x \mapsto x^2$ ist die Funktion $\mathbb{R} \rightarrow \mathbb{R} : x \mapsto 2 \cdot x$.

```
> diff(exp(x), x);
```

$$e^x \quad (2.1.8)$$


MAPLE: Interessant ist, wie Maple sich bei Funktionen verhält, die überall bis auf wenige Ausnahmen differenzierbar sind:

```
> f:=x->piecewise(x<0,x+1,x<1,1,x);
```

```
> f(x);
```

$$\begin{cases} x+1 & x < 0 \\ 1 & x < 1 \\ x & \text{otherwise} \end{cases} \quad (2.1.9)$$

```
> plot(f(x), x=-1..2, thickness=3);
```



```
> diff(f(x),x);
```

$$\left\{ \begin{array}{ll} 1 & x < 0 \\ \text{undefined} & x = 0 \\ 0 & 0 < x < 1 \\ \text{undefined} & x = 1 \\ 1 & 1 < x \end{array} \right.$$

(2.1.10)


oder unter Anwendung des sog. Differentialoperators D:

```
> D(f)(z);
```

$$\left\{ \begin{array}{ll} 1 & z < 0 \\ \text{undefined} & z = 0 \\ 0 & 0 < z < 1 \\ \text{undefined} & z = 1 \\ 1 & 1 < z \end{array} \right.$$

(2.1.11)

```
> plot(diff(f(x),x),x=-1..2,discont=true,thickness=3);
```



```
> f:='f':g:='g':
```


Ein weiteres Beispiel:

```
> diff(abs(x),x);
```

$\text{abs}(1, x)$

(2.1.12)

```
> plot(abs(1,x),x=-1..1,discont=true,thickness=3);
```



DENKANSTOSS: Was haben die obigen Beispiele mit Rechts- bzw. Linksdifferenzierbarkeit zu tun?

Beispiel:


```
> diff((sin(1/x))*x,x);
```

(2.1.13)

$$-\frac{\cos\left(\frac{1}{x}\right)}{x} + \sin\left(\frac{1}{x}\right)$$

(2.1.13)

```
> plot({(sin(1/x))*x},x=-0.1..0.1,thickness=1,numpoints=200);
```



ÜBUNG [02]:

Untersuche mit Hilfe eines gläubigen Vertrauens auf den **limit**-Befehl von MAPLE:

1) Welche der Funktionen $x \mapsto x^j \cdot \sin\left(\frac{1}{x}\right)$ sind in $x = 0$ stetig ergänzbar,

$j = 0, \dots, 4$?

2) Welche dieser Funktionen sind differenzierbar in $x = 0$?

3) Plote die Funktionen und erkläre, wie man die Ergebnisse aus 1) und 2) in den Plots wiederfindet.

Vgl. auch Aufgabe 1 in Abschnitt "Grenzwerte und Stetigkeit".

Ableitungsregeln

MATH und **MAPLE**: Ähnlich wie für Grenzwerte gibt es für Ableitungen Regeln, die MAPLE bekannt sind:

Summenregel:

> diff(f(x)+g(x),x);

$$\frac{d}{dx} f(x) + \frac{d}{dx} g(x) \quad (2.2.1)$$

Für die Ableitung gilt die Produktregel.

> diff(f(x)*g(x),x);

$$\left(\frac{d}{dx} f(x)\right) g(x) + f(x) \left(\frac{d}{dx} g(x)\right) \quad (2.2.2)$$

Daraus folgt die Quotientenregel .

> simplify(diff(f(x)/g(x),x));

$$\frac{\left(\frac{d}{dx} f(x)\right) g(x) - f(x) \left(\frac{d}{dx} g(x)\right)}{g(x)^2} \quad (2.2.3)$$

Einfacher Beweis der Quotientenregel: Es gilt für eine differenzierbare Funktion g :

$$0 = 1' = \left(g \cdot \frac{1}{g}\right)' = g' \cdot \frac{1}{g} + g \cdot \left(\frac{1}{g}\right)'$$

Wenn man nach $\left(\frac{1}{g}\right)'$ umstellt, gilt:

$$\left(\frac{1}{g}\right)' = -\frac{g'}{g^2}.$$

Weiter:

$$\left(f \cdot \frac{1}{g}\right)' = f' \cdot \left(\frac{1}{g}\right) + f \cdot \left(\frac{1}{g}\right)' = \frac{f'}{g} - \frac{f \cdot g'}{g^2} = \frac{f' \cdot g - f \cdot g'}{g^2}.$$

Weiterhin gilt auch hier die Kettenregel:

> **diff(f(g(x)),x);**

$$D(f)(g(x)) \left(\frac{d}{dx} g(x) \right)$$

(2.2.4)

MATH: Die Produktregel ist fundamental.

▼ Differentiation von Polynomen

MATH: Wir hatten bereits die formale Differentiation von Polynomen in einer Variablen behandelt. Wir wollen jetzt Polynomfunktionen im oben definierten Sinne differenzieren:

ÜBUNG [03]:

Zeige: Die Funktion $f: \mathbb{R} \rightarrow \mathbb{R} : x \mapsto x^n$ ist in jedem $x \in \mathbb{R}$ differenzierbar und ihre Ableitung ist $f': \mathbb{R} \rightarrow \mathbb{R} : x \mapsto n \cdot x^{n-1}$.
Hinweis: Binomischer Lehrsatz.

Mit der Summenregel folgt, dass Polynomfunktionen differenzierbar sind und ihre Ableitung der formalen Ableitung der ihnen zugrundeliegenden Polynome entspricht.

▼ Ableitungen von Umkehrfunktionen

MATH: Große Klassen von Funktionen sind differenzierbar (=differenzierbar im Innern ihres Definitionsbereiches), z. B. Polynomfunktionen oder Funktionen, die durch eine konvergente Potenzreihe dargestellt sind. Weiter sind aufgrund der Kettenregel auch Umkehrfunktionen von streng monotonen differenzierbaren Funktionen differenzierbar:

> **exp(log(x));**

$$x$$

(2.4.1)

> **diff(exp(log(x)),x);**

$$1$$

(2.4.2)

Also wegen

> **diff(exp(x),x);**

$$e^x$$

(2.4.3)

folgt nach der Kettenregel $\log'(x) \cdot \exp(\log(x)) = 1$ und somit:

> **diff(log(x),x);**

$$\frac{1}{x}$$

(2.4.4)

DENKANSTOSS: Diskutiere die Definitionsbereiche.

ÜBUNG [04]:

1) Leite die Umkehrfunktion des Tangens, also \arctan , mit Hilfe der Kettenregel nach dem obigen Vorgehen ab.

Hinweis: $\tan := \frac{\sin}{\cos}$. Du brauchst die Definitionsbereiche nicht detailliert diskutieren.


2) Leite mit obiger Idee eine Formel her, welche die Ableitung der Umkehrfunktion ausrechnet. Welche Voraussetzungen sind notwendig?

Der folgende Denkanstoß, dessen Verifikation mit den derzeitigen Hilfsmitteln noch nicht durchführbar ist, soll helfen, Fehlvorstellungen abzubauen:

DENKANSTOSS: Die folgende Funktion ist überall stetig und nirgends differenzierbar:

$$x \mapsto \sum_{n=1}^{\infty} \frac{1}{n^2} \cdot \sin(n^3 \cdot x)$$

> `plot(add(1/n^2*sin(n^3*x),n=1..100),x=-1..1);`



Monotonie, Extrema, Konvexität

[Aufgaben: 2


[> **restart:**
with(plots):

Monotonie

An dem Vorzeichen der Ableitung liest man die Monotonie ab, an dem Vorzeichen der zweiten Ableitung die Konvexitätseigenschaften.

MATH: Stetige Funktionen f auf einem abgeschlossenen Intervall, die im Innern des Intervalles differenzierbar sind, sind auf den Teilintervallen, wo die Ableitung positiv (bzw. negativ) ist, streng monoton steigend (bzw. fallend).

> **plot([x,sin(x)], x=-1..2, color=[red, black]);**



Man vermutet also, dass für $x > 0$ gilt: $x > \sin(x)$. Dies wollen wir mit Hilfe der Ableitung beweisen.

> **diff(x-sin(x),x);**

$$1 - \cos(x)$$

(3.1.1)

ist sicher nicht negativ. Also ist $x \mapsto x - \sin(x)$ monoton steigend. Der Wert bei 0 ist Null, also ist $x - \sin(x) \geq 0$ für alle $x \geq 0$. Aber für jedes x mit $0 < x < 2 \cdot \pi$ ist $1 - \cos(x)$ sogar positiv, so dass $x \mapsto x - \sin(x)$ dort sogar streng monoton wächst. Also haben wir sogar $x - \sin(x) > 0$ für alle $x > 0$.

ÜBUNG [05]:

Zeige $x - \frac{x^3}{6} < \sin(x)$ für $x > 0$.

(Hinweis: Beginne mit der zweiten Ableitung der Differenz.)

Extrema und Konvexität


MATH: Bislang haben wir uns für die monotonen Abschnitte einer differenzierbaren Funktion interessiert, wo also die lineare Approximation $x \mapsto f(x_0) + f'(x_0) \cdot (x - x_0)$ bijektiv ist. Was passiert an den Punkten, wo die lineare Approximation nicht bijektiv, also konstant ist, d. h. $f'(x_0) = 0$? Damit wir uns nicht auf Nebenschauplätzen verlieren, nehmen wir an, dass die Ableitung f' auf dem betrachteten Intervall stetig ist und dass x_0 isolierte Nullstelle von f' ist, d. h. die einzige Nullstelle in einer kleinen offenen Umgebung U von x_0 .


Zwei Fälle sind denkbar:

- 1) f' hat einen Vorzeichenwechsel in x_0
 - 1a) von negativ nach positiv.
 - 1b) von positiv nach negativ.
- 2) f' hat keinen Vorzeichenwechsel in x_0
 - 2a) f' ist nicht negativ in U ,
 - 2b) f' ist nicht positiv in U .

Hier sind Beispiele für die Fälle 1a, 1b, 2a und 2b, wobei wir nicht f' sondern f geplottet haben:

```
> P:=array(1..2,1..2);  
                                     P:= array(1..2, 1..2, [ ])  
                                     (3.2.1)  
=> P[1,1]:=plot(1+x^2,x=-1..1):  
=> P[1,2]:=plot(1-x^2,x=-1..1):  
=> P[2,1]:=plot(1+x^3,x=-1..1):  
=> P[2,2]:=plot(1-x^3,x=-1..1):  
> display(P);
```





MATH: In den Fällen 1a) und 1b) liegt ein lokales Extremum in x_0 vor, genauer ein lokales Minimum im Falle 1a) und ein lokales Maximum im Falle 1b). Setzen wir weitergehend voraus, dass f' monoton steigend im Falle 1a) ist, so ist f auf U **konvex**, d. h.

$$f((1-\lambda) \cdot x + \lambda \cdot y) \leq (1-\lambda) \cdot f(x) + \lambda \cdot f(y)$$

für alle $x < y$ in U , $0 \leq \lambda \leq 1$.

Setzen wir weitergehend voraus, dass f' monoton fallend im Falle 1b) ist, so ist f konkav (d. h. $-f$ konvex). In den Fällen 2a) und 2b) ist x_0 kein Extremum. Man spricht von einem Wendepunkt (Übergang von konkav nach konvex oder von konvex nach konkav), also ein relatives Extremum der Ableitung und damit insbesondere (nicht hinreichend!) Nullstelle der 2. Ableitung, falls die Funktion 2 mal differenzierbar ist.

Nullstellen der Ableitung nennt man allgemein stationäre Punkte. Sie können Extrema oder Wendepunkte sein, wie wir gerade gesehen haben. Sie können aber auch recht wild sein, wenn sie keine isolierten Nullstellen sind, wie das folgende Beispiel zeigt:

> **$f := \sin(1/x) \cdot x^3$** ;

```
f1:=diff(f,x);
```

$$f:=\sin\left(\frac{1}{x}\right)x^3$$

$$f1:= -\cos\left(\frac{1}{x}\right)x + 3\sin\left(\frac{1}{x}\right)x^2$$


(3.2.2)

```
> b:=array(1..1,1..2):
```

```
> b[1,1]:=plot(f,x=-0.01..0.01,numpoints=400):
```

```
> b[1,2]:=plot(f1,x=-0.01..0.01,numpoints=400):
```

```
> display(b);
```



In diesem Fall ist zwar die Ableitung 0 in $x_0 = 0$:

```
> limit(diff(f,x),x=0);
```

0

(3.2.3)

```
> solve(f1=0);
```

$$\frac{1}{\text{RootOf}(-Z - 3 \tan(-Z))}$$

(3.2.4)

Aber x_0 ist keine isolierte Nullstelle der Ableitung. Weiterhin existiert die zweite Ableitung in $x_0 = 0$ nicht.

MATH: Bei einem abgeschlossenen Intervall muss man sich vor dem Trugschluss bewahren, dass die relativen Extrema nur Nullstellen der Ableitung sind. Dies trifft nur für die inneren relativen Extrema zu.

ÜBUNG [06]:

Finde für $f: [-1, 1] \rightarrow \mathbb{R}: x \mapsto x^3 - x + 1$ alle relativen Extrema, die Wendepunkte, die konvexen und konkaven Bereiche und die Nullstellen.

1. Hinweis: Randwerte!
2. Hinweis: Was ist an den Randwerten los?
3. Hinweis: Hast du die Randwerte beachtet?

▼ Funktionenräume, die unter Differentiation abgeschlossen sind

[Aufgaben: 1

> **restart;**
with(LinearAlgebra):

▼ Funktionenräume, die unter Differentiation abgeschlossen sind

Wir betrachten die Menge aller Abbildungen von \mathbb{R} nach \mathbb{R} . Diese wird offensichtlich zu einem \mathbb{R} -Vektorraum durch

- 1.) $+$: $\mathbb{R}^{\mathbb{R}} \times \mathbb{R}^{\mathbb{R}} \rightarrow \mathbb{R}^{\mathbb{R}}: (f, g) \mapsto f + g$, wobei $(f + g)(x) = f(x) + g(x)$
- 2.) \cdot : $\mathbb{R} \times \mathbb{R}^{\mathbb{R}} \rightarrow \mathbb{R}^{\mathbb{R}}: (r, f) \mapsto r \cdot f$, wobei $(r \cdot f)(x) = r \cdot f(x)$

MATH: Die Menge der unendlich oft differenzierbaren (oder glatten) Abbildungen $C^{\infty}(\mathbb{R})$ bildet einen Teilraum von $\mathbb{R}^{\mathbb{R}}$.

Dies lässt sich leicht einsehen, denn möchte man die Summe zweier differenzierbarer Funktionen ableiten, bedient man sich der Summe der Ableitungen. Ähnlich ist die Ableitung eines \mathbb{R} -Vielfachen einer differenzierbaren Funktion ein \mathbb{R} -Vielfaches der Ableitung. Dies bedeutet insbesondere, dass die **Ableitung** eine **\mathbb{R} -lineare Abbildung** ist:

MATH: Die Ableitung

$$\partial: C^{\infty}(\mathbb{R}) \rightarrow C^{\infty}(\mathbb{R})$$

ist ein Vektorraumendomorphismus von $C^{\infty}(\mathbb{R})$.

Da $C^{\infty}(\mathbb{R})$ nicht endlich-dimensional als Vektorraum ist, können wir diese nicht

schön hinschreiben. Wir gehen daher zu einem endlich-dimensionalen Teilraum von $C^\infty(\mathbb{R})$ über.

```
> T:=[sin(x)-2*cos(x)-2*exp(x)+x*exp(x)+2*sin(x)*cos(x)+2-2*x^2, sin(x)-cos(x)-4*exp(x)+2*x*exp(x)+sin(x)*cos(x)+sin(x)^2-4*x^2, -3+cos(x)-exp(x)-x*exp(x)-sin(x)*cos(x)+2*sin(x)^2-x^2, 2*sin(x)-3*cos(x)-5*exp(x)+2*x*exp(x)+5*sin(x)*cos(x)+4*sin(x)^2-5*x^2+1, -x*exp(x)-sin(x)*cos(x)-x^2, -3+exp(x)-3*x*exp(x)+4*sin(x)^2-x^2-2*x, sin(x)-1-8*exp(x)+9*x*exp(x)+3*sin(x)*cos(x)+2*sin(x)^2-6*x^2, 2*sin(x)-4*cos(x)-3*exp(x)-2*x*exp(x)-2*sin(x)*cos(x)-8*sin(x)^2-x^2+3*x+7, sin(x)-3*cos(x)-exp(x)+7+sin(x)*cos(x)-3*sin(x)^2-4*x^2+x];
```

$$T := [\sin(x) - 2 \cos(x) - 2 e^x + x e^x + 2 \sin(x) \cos(x) + 2 - 2 x^2, \sin(x) - \cos(x) - 4 e^x + 2 x e^x + \sin(x) \cos(x) + \sin(x)^2 - 4 x^2, -3 + \cos(x) - e^x - x e^x - \sin(x) \cos(x) + 2 \sin(x)^2 - x^2, 2 \sin(x) - 3 \cos(x) - 5 e^x + 2 x e^x + 5 \sin(x) \cos(x) + 4 \sin(x)^2 - 5 x^2 + 1, -x e^x - \sin(x) \cos(x) - x^2, -3 + e^x - 3 x e^x + 4 \sin(x)^2 - x^2 - 2 x, \sin(x) - 1 - 8 e^x + 9 x e^x + 3 \sin(x) \cos(x) + 2 \sin(x)^2 - 6 x^2, 2 \sin(x) - 4 \cos(x) - 3 e^x - 2 x e^x - 2 \sin(x) \cos(x) - 8 \sin(x)^2 - x^2 + 3 x + 7, \sin(x) - 3 \cos(x) - e^x + 7 + \sin(x) \cos(x) - 3 \sin(x)^2 - 4 x^2 + x] \quad (4.1.1)$$

Wir möchten zunächst die lineare Unabhängigkeit prüfen. Dazu bedienen wir uns der folgenden einfachen Tatsache aus der linearen Algebra:

Seien V, W K -Vektorräume, $f: V \rightarrow W$ ein Vektorraumhomomorphismus, $v_1, \dots, v_k \in V$ und $a_1, \dots, a_k \in K$. Dann gilt:

$$\sum_{i=1}^k a_i v_i = 0 \Rightarrow f\left(\sum_{i=1}^k a_i v_i\right) = \sum_{i=1}^k a_i f(v_i) = 0.$$

Mit anderen Worten, jede lineare Abhängigkeit der v_i ist auch eine lineare Abhängigkeit der $f(v_i)$. Umgekehrt gilt: Sind die $f(v_i)$ linear unabhängig, dann sind auch die v_i linear unabhängig.

Möchte man lineare Abhängigkeiten der v_i finden kann man wie folgt vorgehen: Man findet zunächst lineare Abhängigkeiten der $f(v_i)$ und testet dann, ob diese schon lineare Abhängigkeiten der v_i sind (dieser Test kann natürlich weggelassen werden, wenn f ein Monomorphismus ist).

Wir entwickeln nun die Funktionen in Potenzreihen, schneiden ab Grad k ab und betrachten die Koeffizientenvektoren. Die Abbildung der Funktion auf diesen Koeffizientenvektor ist ein Homomorphismus von $C^\infty(\mathbb{R}) \rightarrow \mathbb{R}^k$. Dann wenden wir die eben zitierte Methode an.

Das folgende Programm ist eine leichte Verbesserung des Programms aus dem Abschnitt "Formale Potenzreihen".


```

> Entw:=proc(f,n::posint,x::name)
  local g,i,L,y;
  L:=NULL;
  g:=f;
  for i from 0 to n+1 do
    y:=limit(g,x=0);
    L:=L,y;
    g:=simplify((g-y)/x);
  end do;
  return add(L[i]*x^(i-1),i=1..n);
end proc:

```

```

> TE:=Transpose(Matrix(map(f->[seq(coeff(f,x,i), i=0..nops(T)
-1)], map(Entw, T, nops(T), x))));

```

$$TE := \begin{bmatrix}
-2, -5, -3, -7, 0, -2, -9, 0, 3 \\
2, 0, -3, 4, -2, -4, 5, -2, 2 \\
-1, -\frac{5}{2}, -1, 0, -2, \frac{1}{2}, 1, -\frac{21}{2}, -6 \\
-\frac{4}{3}, -\frac{1}{2}, 0, -\frac{7}{2}, \frac{1}{6}, -\frac{4}{3}, 1, -\frac{1}{2}, -1 \\
0, -\frac{5}{24}, -\frac{5}{6}, -\frac{4}{3}, -\frac{1}{6}, -\frac{43}{24}, \frac{1}{2}, \frac{49}{24}, \frac{5}{6} \\
\left[\frac{3}{10}, \frac{23}{120}, -\frac{11}{60}, \frac{29}{40}, -\frac{7}{40}, -\frac{7}{60}, \frac{43}{60}, -\frac{43}{120}, \frac{2}{15}\right] \\
\left[\frac{1}{120}, \frac{41}{720}, \frac{7}{90}, \frac{23}{120}, -\frac{1}{120}, \frac{37}{240}, \frac{11}{72}, -\frac{89}{240}, -\frac{47}{360}\right] \\
\left[-\frac{31}{1260}, -\frac{11}{1008}, \frac{1}{90}, -\frac{313}{5040}, \frac{19}{1680}, -\frac{1}{252}, -\frac{23}{840}, \frac{109}{5040}, -\frac{11}{840}\right] \\
\left[\frac{1}{10080}, -\frac{13}{4480}, -\frac{11}{1680}, -\frac{1}{80}, -\frac{1}{5040}, -\frac{107}{8064}, -\frac{1}{210}, \frac{143}{5760}, \frac{19}{2016}\right]
\end{bmatrix} \tag{4.1.2}$$

In den Spalten stehen nun jeweils die ersten 9 Koeffizienten der gegebenen Funktionen. Wir sehen, dass diese Matrix vollen Rang hat und somit T linear unabhängig in $C^\infty(\mathbb{R})$ ist:

```

> Rank(TE),nops(T);

```

9, 9

(4.1.3)

ÜBUNG [07]:

Sei $V \leq C^\infty(\mathbb{R})$ erzeugt von T .

1.) Begründe, warum die Elemente von T linear unabhängig sind, also einen Raum der Dimension 9 erzeugen.

- 2.) Zeige, dass V unter der Differentiation abgeschlossen ist, also dass $\partial: V \rightarrow V$ eine wohldefinierte Abbildung ist.
- 3.) Bestimme die Matrix Md von ∂ bezüglich der Basis T .
- 4.) Schränke die Abbildung ∂ auf den von $(\sin(x), \cos(x), x \cdot \exp(x), \exp(x))$ erzeugten Teilraum ein, indem du eine dafür geeignete Basis wählst, die Ableitung als Matrix $Md\partial$ bezüglich dieser Basis schreibst und anschließend auf eine Teilmatrix einschränkst.
- 5.) Die Matrix aus 4.) ist invertierbar. Wieso? Berechne ihre Inverse und bestimme das Bild von $x \cdot \exp(x)$.